

9/10.5

Calcul du temps nécessaire à la réalisation d'un parcours

Le programme présenté dans ce chapitre est destiné aux sportifs, et plus particulièrement à ceux qui pratiquent la marche à pied ou le cyclo-tourisme.

Il permet en effet de calculer et d'afficher sous la forme d'un tableau les diverses étapes d'un parcours, et, en fonction d'une moyenne horaire prévue, les heures de passage à chaque étape.

Le programme est écrit en Basic. Si vous le désirez, il s'adapte facilement en Turbo-Pascal.

COMMENT UTILISER LE PROGRAMME

Le listing du programme est le suivant :

```

1000 ' *****
1010 ' *           P A R C O U R S           *
1020 ' *****
1030 '
1040 ' -----
1050 ' Affichage ecran de presentation
1060 ' -----
1070 '
1080 MODE 1
1090 LOCATE 12,10
1100 PRINT"R A N D O N N E E "
1110 LOCATE 10,25
1120 PRINT"Appuyez sur une touche ..."
1130 LOCATE 11,8
1140 PRINT CHR$(150);STRING$(17,CHR$(154));CHR$(156)
1150 LOCATE 11,10
1160 z$=CHR$(149)
1170 PRINT z$
1180 LOCATE 11,9
1190 PRINT z$
1200 LOCATE 11,11
1210 PRINT z$
1220 LOCATE 29,10
1230 PRINT z$
1240 LOCATE 29,9
1250 PRINT z$
1260 LOCATE 29,11
1270 PRINT z$
1280 LOCATE 11,12
1290 PRINT CHR$(147);STRING$(17,CHR$(154));CHR$(153)
1300 '
1310 ' -----
1320 ' Avertisseur sonore
1330 ' -----
1340 '
1350 MX=40
1360 FOR i=1 TO 2
1370   SOUND 3,600,36,6
1380   SOUND 5,500,48,8
1390   SOUND 3,400,60,8
1400   SOUND 5,300,72,6
1410   GOSUB 2480
1420 NEXT i
1430 '
1440 ' -----
1450 ' Attente de la frappe d'une touche
1460 ' -----
1470 '
1480 a$=INKEY$:IF a$="" THEN 1480
1490 MODE 2

```

```

1500 '
1510 ' -----
1520 ' Donnees d'affichage
1530 ' -----
1540 '
1550 a$="*****"
*****"
1560 b$="          *          *          *
*          *"
1570 c$="          * LOCALITE          * KM          * KM
* HORAIRE          *"
1580 d$="          * L$(I)          * KMP(I) * KMT(I)
* H(I)hM(I)          *"
1590 e$="          *          * partiels* totaux
*          *"
1600 '
1610 LOCATE 6,7
1620 INPUT "Localite de depart L1.....";L$(1)
1630 LOCATE 6,9
1640 INPUT "Premiere etape L2.....";L$(2)
1650 LOCATE 6,11
1660 INPUT "Deuxieme etape L3.....";L$(3)
1670 LOCATE 6,13
1680 INPUT "Troisieme etape L4.....";L$(4)
1690 LOCATE 6,15
1700 INPUT "Quatrieme etape L5.....";L$(5)
1710 LOCATE 6,17
1720 INPUT "Arrivee localite L6.....";L$(6)
1730 mx=80
1740 GOSUB 2480
1750 GOSUB 2480
1760 LOCATE 50,8
1770 INPUT "Distance... L1 a L2.....";KMP(2)
1780 LOCATE 50,10
1790 INPUT "Distance... L2 a L3.....";KMP(3)
1800 LOCATE 50,12
1810 INPUT "Distance... L3 a L4.....";KMP(4)
1820 LOCATE 50,14
1830 INPUT "Distance... L4 a L5.....";KMP(5)
1840 LOCATE 50,16
1850 INPUT "Distance... L5 a L6.....";KMP(6)
1860 MX=80
1870 GOSUB 2480
1880 GOSUB 2480
1890 CLS
1900 LOCATE 12,9
1910 INPUT "          Quelle est votre vitesse . . . . . ";V
1920 PRINT#1:PRINT#1
1930 LOCATE 12,12
1940 INPUT "          Quelle est votre heure de depart . . ";HD

```

```

1950 PRINT#1:PRINT#1
1960 LOCATE 12,15
1970 INPUT "      Et combien de minutes. . . . . ";MD
1980 GOSUB 2480
1990 GOSUB 2480
2000 KMP(1)=0
2010 CLS
2020 LOCATE 16,4
2030 PRINT"Votre vitesse est:                ";V;"KM/H"
2040 PRINT
2050 LOCATE 16,6
2060 PRINT"Votre horaire de depart est:      ";HD;"h";MD;"m"

2070 I=1
2080 FOR I=1 TO 6
2090   LOCATE 16,6
2100   KMT(I)=KMP(I)+KMT(I-1)
2110   H(I)=(KMT(I)/V)+HD
2120   M(I)=(KMT(I)/V*60)+MD
2130 NEXT
2140 PRINT
2150 PRINT
2160 PRINT a$
2170 PRINT b$
2180 PRINT c$
2190 PRINT e$
2200 PRINT a$
2210 FOR i=1 TO 6
2220   GOSUB 2360
2230   PRINT TAB(14);"*";TAB(18);L$(I);TAB(32);
2240   PRINT "*" ;TAB (37);KMP(I);TAB(42);"*";
2250   PRINT TAB(47);KMT(I);TAB(53);"*";TAB(55);
2260   PRINT H(I);TAB(59);"h";TAB(61);M(I);
2270   PRINT TAB(65);"m";TAB(67);"*"
2280 NEXT
2290 PRINT A$
2300 LOCATE 30,22
2310 INPUT"Une autre estimation (O/N) : ";bis$
2320 IF UPPER$(bis$)="O" THEN RUN
2330 END
2340 '
2350 ' -----
2360 ' Sous-programme de conversion heure minute
2370 ' -----
2380 '
2390 h=H(i)-INT(H(i))
2400 H(I)=INT(H(I))
2410 M(I)=INT(h*60)+MD
2420 RETURN
2430 '

```

```

2440 ' -----
2450 ' Sous-programme d'affichage
2460 ' -----
2470 '
2480 FOR X=MX-1 TO 1 STEP -2
2490   LOCATE X,20
2500   PRINT CHR$(251)
2510 NEXT X
2520 '
2530 FOR N=1 TO 500
2540 NEXT N
2550 '
2560 FOR X=1 TO MX-1 STEP 2
2570   LOCATE X,20
2580   PRINT CHR$(250)
2590 NEXT X
2600 '
2610 FOR n=1 TO 20
2620 NEXT n
2630 RETURN

```

Exécuter le programme. Il vous demande d'entrer :

- la localité de départ L1 ;
- les étapes prévues (L2 à L6) ;
- les distances qui séparent la localité de départ et la première étape et les étapes entre elles ;
- la vitesse de croisière ;
- l'heure de départ.

Il établit ensuite un tableau qui regroupe les heures de passage prévues.


Exemple :

Votre vitesse est : 19 km/h
 Votre horaire de départ est : 6 h 15 mn

Localité	Km partiels	Km totaux	Horaire
Donge	0	0	6 h 15
Linac	54	54	8 h 09
Bastide	28	82	10 h 33
Claribe	63	145	13 h 52
St-Sernain	30	175	15 h 27
Levis	72	247	19 h 15

LE PROGRAMME EN DÉTAIL

La logique du programme apparaît dans l'ordinogramme suivant :



Les lignes 1000 à 1290 affichent le titre du programme. Ce titre est encadré à l'aide de caractères graphiques.

Un marcheur en pleine action est représenté par les caractères de codes ASCII 251 et 250. Son affichage se fait dans un sous-programme qui occupe les lignes 2440 et 2630. Une petite musique cadence son évolution :

```
1360 FOR i = 1 TO 2
1370  SOUND 3, 600, 36, 6
1380  SOUND 3, 500, 48, 8
1390  SOUND 3, 400, 60, 8
1400  SOUND 5, 300, 72, 6
1410  GOSUB 2480
1420 NEXT i
```

Le programme se met alors en attente de la frappe d'une touche du clavier lignes 1480 à 1490.

L'en-tête du tableau résultat est calculé et stocké dans les variables chaînes **a\$** et **e\$** entre les lignes 1550 et 1590.

Le programme demande ensuite d'entrer les diverses données textes et numériques entre les lignes 1610 et 2060.

La suite du programme convertit les données numériques entrées en un horaire de passage aux étapes prévues et affiche les résultats dans un tableau.

Les dernières lignes du programme (lignes 2350 à 2630) contiennent deux sous-programmes :

- conversion d'une durée en heures et minutes ;
- affichage du marcheur entre deux étapes du programme.

